

Practica 9. Uso del control MSFlexGrid

Objetivos:

Al finalizar esta práctica el estudiante será capaz de:

1. Usar el control MSFlexGrid para mostrar y procesar una matriz.
2. Usar los Arreglos para guardar información
3. Procesar una encuesta para generar estadísticas básicas y calcular el modelo de regresión
4. Crear la siguiente aplicación:

Se desea que Usted realice un programa que procese la encuesta y:

- a) Calcule las Estadísticas Básicas (Media, Varianza, Mínimo y Máximo)
- b) Estime los parámetros de la Recta de Regresión entre el Peso y la Estatura

Encuesta

- Edad: _____
- Peso: _____
- Estatura: _____

Antes de comenzar

Realizar el Análisis E-P-S de la Aplicación Encuesta.

Entrada.

Dato	Identificador	Tipo	Restricciones
Edad del entrevistado	Edad	Real	>15
Peso del entrevistado	Peso	Real	>0
Estatura del entrevistado	Estatura	real	>0

Tabla 9-1. Análisis EPS de la aplicación "Encuesta".

Proceso

- Cada vez que se lee un dato se incrementa el número de entrevistados (N)
- Para calcular el promedio se llevan tres acumuladores para sumar los datos de cada uno de los entrevistados

Media :

$$SEdad = \sum_{i=1}^n Edad(i)$$

$$SPeso = \sum_{i=1}^n Peso(i)$$

$$SEstatura = \sum_{i=1}^n Estatura(i)$$

$$MEdad = SEdad / N$$

$$MPeso = SPeso / N$$

$$MEstatura = SEstatura / N$$

Varianza :

$$S2Edad = \sum_{i=1}^n Edad(i)^2$$

$$S2Peso = \sum_{i=1}^n Peso(i)^2$$

$$S2Estatura = \sum_{i=1}^n Estatura(i)^2$$

$$VEdad = \frac{S2Edad - N * MEdad}{n - 1}$$

$$VPeso = \frac{S2Peso - N * MPeso}{n - 1}$$

$$VEstatura = \frac{S2Estatura - N * MEstatura}{n - 1}$$

Para el mínimo (y el máximo) se recorre el arreglo y se va comparando cada valor con el mínimo (máximo) y en caso de encontrar otro que sea menor (mayor) se cambia.

Re gresión

$$SPesoEst = \sum_{i=1}^n Estatura(i) * Peso(i)$$

$$Beta1 = \frac{N * SPesoEst - SEstatura * SPeso}{N * S2Estatura - SEstatura^2}$$

$$Beta0 = MPeso - Beta1 * MEstatura$$

Salida

La media, la varianza, el mínimo y el máximo para cada variable

Los parámetros de la recta de regresión

Para la interfaz de usuario se usaran los siguientes controles:

- 3 etiquetas (label)
- 3 cajas de texto (TextBox)
- 4 botones de comando (CommandButton)
- 1 Cuadrícula Flexible (MSFlexGrid)

Actividades a realizar:

1. Ejecute el Visual Basic 6.0, seleccione “Exe estándar” en el cuadro de dialogo “Nuevo Proyecto”.
2. Establezca las siguientes propiedades al formulario:

Propiedad	Valor
Caption	Practica 9. Ms FlexGrid
Forecolor	A su gusto
Height	6200
StartPosition	2 - CenterScreen
Width	8600

Tabla 9-2. Propiedades del formulario 1 de la aplicación “Encuesta”.

Este formulario contiene en la parte izquierda los cuadros de texto para recoger los datos y guardarlos en tres arreglos (Edad, Peso y Estatura) y en la parte derecha el control MsFlexGrid que se utilizará para mostrar los datos que se han ido leyendo.

3. Coloque los siguientes controles sobre el formulario aproximadamente como se muestra en la figura 9-1:


- 3 etiquetas (Label)
- 3 cuadros de texto (TextBox)
- 4 botones de comando (CommandButton)
- 1 cuadrícula flexible de datos (MsFlexGrid-)



Figura 9-1. Formulario 1 de la aplicación “Encuesta”.

4. En la tabla 9-3 se presenta las propiedades de los controles que se agregaron al formulario:

Control	Propiedad	Valor
Etiqueta 1	(Nombre)	Label1
	Caption	Edad:
	Width, Height, Top y Left	2175, 375, 960, 240
	Alignment	1 – Right Justify
	Font	Ms Sans Serif, Normal, 14

Etiqueta 2	(Nombre)	Label2
	Caption	Peso:
	Width, Height, Top y Left	2175, 375, 1800, 240
	Alignment	1 – Right Justify
	Font	Ms Sans Serif, Normal, 14
Etiqueta 3	(Nombre)	Label3
	Caption	Estatura:
	Width, Height, Top y Left	2175, 375, 2640, 240
	Alignment	1 – Right Justify
	Font	Ms Sans Serif, Normal, 14
Cuadro de texto 1	(Nombre)	Text1
	Text	
	Width, Height, Top y Left	1250, 625, 840, 2600
	Font	Arial, Normal, 10
Cuadro de texto 2	(Nombre)	Text2
	Text	1250, 625, 1680, 2600
	Width, Height, Top y Left	
	Font	Arial, Normal, 10
Cuadro de texto 3	(Nombre)	Text3
	Text	
	Width, Height, Top y Left	1250, 625, 2520, 2600
	Font	Arial, Normal, 10
Botón 1	(Nombre)	Command1
	Caption	&Leer Dato
	Width, Height, Top y Left	1650, 500, 5000, 200
	Font	Arial, Negrita, 10
Botón 2	(Nombre)	Command2
	Caption	&Procesar
	Width, Height, Top y Left	1650, 500, 5000, 2200
	Font	Arial, Negrita, 14
Botón 3	(Nombre)	Command3
	Caption	&Inicializar
	Width, Height, Top y Left	1650, 500, 5000, 4200
	Font	Arial, Negrita, 14
Botón 4	(Nombre)	Command4
	Caption	&Salir
	Width, Height, Top y Left	1650, 500, 5000, 6200
	Font	Arial, Negrita, 14

Cuadrícula 1	(Nombre)	MsFlexGrid1
	Width, Height, Top y Left	4200, 4800, 100, 4200
	Font	Arial, Negrita, 14

Tabla 9-3. Propiedades de los controles del formulario1 de la aplicación "Encuesta".

5. Inserte el Código del Formulario 1. Este código contiene, en primer lugar la inicialización de los controles de lectura, después una función para validar datos y luego el código de los 4 botones de comando.

Option Explicit

Public Sub Inicializar_controles()

Text1.Text = ""

Text2.Text = ""

Text3.Text = ""

End Sub

' Función para validar los datos

Public Function Validar() As Boolean

Dim X As Single

Validar = True

If Not IsNumeric(Text1.Text) Then

MsgBox "Error... La Edad es numerica", vbOKOnly, "Error"

Text1.SetFocus

Validar = False

Exit Function

Else

X = Text1.Text

If X < 15 Then

MsgBox "Error... La Edad es mayor o igual a 15", vbOKOnly, "Error"

Text1.SetFocus

Validar = False

Exit Function

End If

End If

If Not IsNumeric(Text2.Text) Then

MsgBox "Error... El Peso es numerico", vbOKOnly, "Error"

Text2.SetFocus

Validar = False

Exit Function

Else

X = Text2.Text

If X < 40 Then

MsgBox "Error... El Peso es mayor o igual a 40", vbOKOnly, "Error"

Text2.SetFocus

Validar = False

Exit Function

End If

End If

```
If Not IsNumeric(Text3.Text) Then
    MsgBox "Error... La Estatura es numerica", vbOKOnly, "Error"
    Text3.SetFocus
    Validar = False
    Exit Function
Else
    X = Text2.Text
    If X < 1.4 Then
        MsgBox "Error... La estatura es mayor o igual a 1.40", vbOKOnly,
"Error"
        Text3.SetFocus
        Validar = False
        Exit Function
    End If
End If

End Function

'
' Leer los datos y guardarlos en los arreglos

Private Sub Command1_Click()

    If Validar() Then
        ' Pasar los datos de cuadro de texto para los ARREGLOS
        N = N + 1
        If N = 1 Then
            Command3.Enabled = True
        Else
            If N = 2 Then
                Command2.Enabled = True
            End If
        End If
        Edad(N) = Text1.Text
        Peso(N) = Text2.Text
        Estatura(N) = Text3.Text

        ' Mostrar los datos en el MSFlexGrid
        MSFlexGrid1.Rows = N + 1
        MSFlexGrid1.Row = N
        MSFlexGrid1.Col = 0
        MSFlexGrid1.Text = N
        MSFlexGrid1.Col = 1
        MSFlexGrid1.Text = Edad(N)
        MSFlexGrid1.Col = 2
        MSFlexGrid1.Text = Peso(N)
        MSFlexGrid1.Col = 3
        MSFlexGrid1.Text = Estatura(N)

        Call Inicializar_controles
    End If

End Sub

'
' Procesar
```

```
Private Sub Command2_Click()
    Call Estadisticas
    Form1.Enabled = False
    Form2.Visible = True
End Sub

'
' Inicializar los datos
Private Sub Command3_Click()
    N = 0
    MSFlexGrid1.Rows = 1
    Command2.Enabled = False
    Command3.Enabled = False
End Sub

'
' Salir

Private Sub Command4_Click()
    End
End Sub

Private Sub Form_Load()
    Dim i As Integer

    ' Fijar número de filas y columnas
    MSFlexGrid1.Cols = 4
    MSFlexGrid1.Rows = 2

    ' Fijar número de filas y columnas fijas o encabezados
    MSFlexGrid1.FixedCols = 1
    MSFlexGrid1.FixedRows = 1

    ' Tamaño (ancho) total de cada columna
    MSFlexGrid1.Width = 4200
    MSFlexGrid1.ColWidth(0) = 400
    For i = 1 To 3
        MSFlexGrid1.ColWidth(i) = 1200
        MSFlexGrid1.ColAlignment(i) = 3      ' Alineación
    Next i

    ' Encabezados de cada columna
    MSFlexGrid1.Row = 0
    MSFlexGrid1.Col = 0
    MSFlexGrid1.Text = " N° "
    MSFlexGrid1.Col = 1
    MSFlexGrid1.Text = " Edad "
    MSFlexGrid1.Col = 2
    MSFlexGrid1.Text = " Peso "
    MSFlexGrid1.Col = 3
    MSFlexGrid1.Text = " Estatura "

    Call Inicializar_controles

    Command2.Enabled = False
    Command3.Enabled = False
```



```
N = 0
End Sub
```

6. Almacene la aplicación tal y como se ha hecho en las prácticas anteriores.
7. Inserte un nuevo FORMULARIO (Form2). En este formulario se presentan los resultados obtenidos del programa (ver figura 9-2)
8. Establezca las siguientes propiedades al formulario 2:

Propiedad	Valor
Caption	Resultados de la Encuesta
Forecolor	A su gusto
Height	6400
StartPosition	2 - CenterScreen
Width	8500

Tabla 9-4. Propiedades del formulario 2 de la aplicación "Encuesta".

Form2

Estadísticos Básicos y Regresión Lineal :

Variable	Mínimo	Máximo	Media	Varianza
Edad	19	28	22,8	13,70004
Peso	52	72	62,6	69,80024
Estatura	1,6	1,82	1,72	8,949826E-03

Recta de Regresión
 Peso = 65,9248 * Estatura + -50,79066

Regresar

Figura 9-2. Formulario 2 de la aplicación "Encuesta".

9. Coloque sobre este formulario una etiqueta para mostrar los resultados y un Botón de ordenes para regresar con las siguientes propiedades:

Control	Propiedad	Valor
Etiqueta 1	(Nombre)	Label1
	Caption	Estadísticos Básicos y Regresión Lineal
	Width, Height, Top y Left	7500, 495, 120, 400
	Alignment	2 – Center
	Font	Arial, Negrita, 18
Etiqueta 2	(Nombre)	Label2
	Caption	Resultados:
	Width, Height, Top y Left	8100, 4500, 750, 120
	Alignment	0 – Left Justify
	Font	Courier New, Normal, 9
Botón 1	(Nombre)	Command1
	Caption	&Regresar
	Width, Height, Top y Left	2000, 400, 5400, 6000
	Font	Arial, Negrita, 10

Tabla 9-5. Propiedades de los controles del formulario 2 de la aplicación “Encuesta”.

Observe que la fuente de la etiqueta 2 es la Courier New, la cual es igualmente espaciada, lo que permite que la salida se vea en forma tabular.

10. Coloque el Código del Formulario 2 para activar el formulario 1 una vez que se desee salir de los resultados.

```
Private Sub Command1_Click()  
    Form1.Enabled = True  
    Form2.Visible = False  
End Sub
```

11. Inserte un Módulo.

Aquí se declaran las variables globales: número de personas y los arreglos Edad, Peso y Estatura; y el procedimiento Estadísticas (botón procesar),

12. Coloque en el Módulo el siguiente Código:

```
Option Explicit  
  
Global N As Integer      ' Número de personas  
  
Global Edad(100) As Single
```

```

Global Peso(100) As Single
Global Estatura(100) As Single

Public Sub Estadisticas()
    Dim MEdad As Single, MPeso As Single, MEstatura As Single
    Dim VEdad As Single, VPeso As Single, VEstatura As Single
    Dim S2Edad As Single, S2Peso As Single, S2Estatura As Single
    Dim MiEdad As Single, MiPeso As Single, MiEstatura As Single
    Dim MaEdad As Single, MaPeso As Single, MaEstatura As Single
    Dim I As Integer
    Dim SPesEst As Single
    Dim Beta1 As Single, Beta0 As Single
    Dim Cad As String, Ent As String

    Ent = Chr(10) & Chr(13)

    '
    ' Inicializar las sumas, Mínimos y Máximos
    MEdad = 0
    MPeso = 0
    MEstatura = 0
    SPesEst = 0
    S2Edad = 0
    S2Peso = 0
    S2Estatura = 0
    MiEdad = 1E+16
    MiPeso = 1E+16
    MiEstatura = 1E+16
    MaEdad = -10
    MaPeso = -10
    MaEstatura = -10

    For I = 1 To N
        ' Sumas
        MEdad = MEdad + Edad(I)
        MPeso = MPeso + Peso(I)
        MEstatura = MEstatura + Estatura(I)
        SPesEst = SPesEst + Peso(I) * Estatura(I)

        ' Sumas al cuadrado
        S2Edad = S2Edad + Edad(I) ^ 2
        S2Peso = S2Peso + Peso(I) ^ 2
        S2Estatura = S2Estatura + Estatura(I) ^ 2

        ' Mínimos y Máximos
        If MiEdad > Edad(I) Then
            MiEdad = Edad(I)
        End If
        If MiPeso > Peso(I) Then
            MiPeso = Peso(I)
        End If
        If MiEstatura > Estatura(I) Then
            MiEstatura = Estatura(I)
        End If
        If MaEdad < Edad(I) Then

```

```

        MaEdad = Edad(I)
    End If
    If MaPeso < Peso(I) Then
        MaPeso = Peso(I)
    End If
    If MaEstatura < Estatura(I) Then
        MaEstatura = Estatura(I)
    End If
Next I

'
'Coeficientes de Regresión
Beta1 = (N * SPesEst - MEstatura * MPeso) / (N * S2Estatura - MEstatura ^ 2)
Beta0 = MPeso / N - Beta1 * MEstatura / N

'
' Medias y Varianzas
MEdad = MEdad / N
VEdad = (S2Edad - N * MEdad ^ 2) / (N - 1)
MPeso = MPeso / N
VPeso = (S2Peso - N * MPeso ^ 2) / (N - 1)
MEstatura = MEstatura / N
VEstatura = (S2Estatura - N * MEstatura ^ 2) / (N - 1)

' Aqui se muestra los resultados en la etiqueta 2 del formulario 2

Cad = "; Variable ; Mínimo      ; Máximo      ; Media      ; Varianza      ;" & Ent
Cad = Cad & "-----+-----+-----+-----+-----" &
Ent
Cad = Cad & "; Edad          ;" & Format(MiEdad, "#####") & " ;" &
Format(MaEdad, "#####") & " ;"
Cad = Cad & Format(MEdad, "#####") & " ;" & Format(VEdad, "#####")
& " ;" & Ent
Cad = Cad & "; Peso          ;" & Format(MiPeso, "#####") & " ;" &
Format(MaPeso, "#####") & " ;"
Cad = Cad & Format(MPeso, "#####") & " ;" & Format(VPeso, "#####")
& " ;" & Ent
Cad = Cad & "; Estatura ;" & Format(MiEstatura, "#####") & " ;" &
Format(MaEstatura, "#####") & " ;"
Cad = Cad & Format(MEstatura, "#####") & " ;" & Format(VEstatura,
"#####") & " ;" & Ent
Cad = Cad & "-----+-----+-----+-----+-----" &
Ent
Form2.Label2.Caption = Cad & Ent & Ent

Cad = "Recta de Regresión" & Ent
Cad = Cad & "Peso = " & Format(Beta1, "#####") & " * Estatura + " &
Format(Beta0, "#####")

Form2.Label2.Caption = Form2.Label2.Caption & Cad & Ent
End Sub

```

13. Almacene de nuevo la aplicación

14. Ejecute la aplicación y verifique su funcionamiento

